
Abstrak 

Sistem hidroponik merupakan metode bercocok tanam tanpa tanah yang mengandalkan 

larutan nutrisi sebagai media utama. Namun, pengelolaan nutrisi yang dilakukan secara manual 

sering kali menyebabkan inefisiensi serta ketidakakuratan dalam pengaturan parameter penting 

seperti TDS, pH, dan suhu. Oleh karena itu, tugas akhir ini bertujuan merancang dan membangun 

sistem kontrol nutrisi hidroponik otomatis berbasis Internet of Things (IoT). Sistem menggunakan dua 

mikrokontroler ESP32 yang berfungsi sebagai unit pemantauan dan pengendali. ESP32 pertama 

membaca data dari sensor TDS, pH, suhu, kelembaban, intensitas cahaya, dan tinggi permukaan air, 

kemudian mengirimkan data ke ESP32 kedua melalui protokol ESP-NOW. ESP32 kedua 

mengendalikan pompa dan solenoid valve secara otomatis berdasarkan parameter yang diterima. 

Sistem juga mendukung kontrol manual melalui Telegram Bot yang memungkinkan pengguna 

mengontrol perangkat dari jarak jauh dengan perintah berbasis teks. Pengujian sistem dilakukan 

secara bertahap, mencakup pengujian sensor, komunikasi antar mikrokontroler, dan integrasi penuh 

dengan Telegram. Hasil menunjukkan bahwa sistem mampu merespons perubahan kondisi 

lingkungan secara real-time dan memberikan notifikasi saat terjadi anomali. Sistem ini dinilai efisien, 

fleksibel, dan cocok untuk skala rumah tangga maupun urban farming. Dengan pendekatan ini, 

diharapkan masyarakat dapat lebih mudah mengelola sistem hidroponik secara modern dan 

berkelanjutan. 

Kata Kunci :   ESP32, Hidroponik Otomatis, Telegram Bot, Monitoring Nutris 

ABSTRACT  

Hydroponics is a method of growing plants without soil, relying on nutrient solutions as 

the primary medium. However, manual nutrient management often results in inefficiencies 

and inaccuracies in maintaining crucial parameters such as TDS, pH, and temperature. 

Therefore, this final project aims to design and develop an automatic nutrient control 

system for hydroponics based on the Internet of Things (IoT). The system utilizes two ESP32 

microcontrollers functioning as monitoring and control units. The first ESP32 reads data 

from TDS, pH, temperature, humidity, light intensity, and water level sensors, then sends 

the data to the second ESP32 using the ESP-NOW protocol. The second ESP32 

automatically controls pumps and solenoid valves based on the received parameters. 

Additionally, the system supports manual control through a Telegram Bot, allowing users 

to operate the system remotely via text-based commands. System testing was conducted in 

stages, including sensor testing, microcontroller communication, and full integration with 

Telegram. The results showed that the system could respond to environmental changes in 

real time and send notifications when anomalies occurred. This system is considered 

efficient, flexible, and suitable for household-scale or urban farming applications. With this 

approach, it is expected that the community can more easily manage hydroponic systems 

in a modern and sustainable manner. 

Keywords : IoT, ESP32,  Hidroponik Otomatis, Telegram Bot, Monitoring Nutrisi 

Bab I 

Pendahuluan 

1.1. Latar Belakang 

  Dalam era pertanian modern, hidroponik menjadi solusi unggulan untuk 

budidaya tanaman tanpa tanah, khususnya di daerah urban yang memiliki 

keterbatasan lahan. Sistem ini mampu menghemat penggunaan lahan dan air secara 

signifikan, sehingga cocok untuk diterapkan di lingkungan perkotaan (Rahman et 

al., 2020). Selain itu, hidroponik memungkinkan kontrol yang lebih presisi terhadap 

nutrisi tanaman. Namun, dalam praktiknya, sebagian besar sistem hidroponik masih 

dioperasikan secara manual, terutama dalam pemberian nutrisi, yang berisiko 

terhadap ketidaktepatan dosis dan waktu pemupukan (Putra & Yusuf, 2022). 

  Penerapan teknologi berbasis mikrokontroler seperti ESP32 menjadi solusi 

potensial untuk mengotomatiskan sistem hidroponik. Dengan sistem ini, berbagai 

parameter lingkungan seperti suhu, kelembapan, TDS, intensitas cahaya, dan 

ketinggian air dapat dimonitor secara real-time. Selain itu, integrasi dua ESP32 satu 

untuk pembacaan sensor dan satu untuk pengontrol aktuator dengan komunikasi 

ESP-NOW memungkinkan pengoperasian yang efisien dan stabil. Sistem ini juga 

dilengkapi dengan kontrol manual berbasis Telegram, memberikan fleksibilitas 

bagi pengguna untuk mengontrol sistem dari jarak jauh (Kumar & Singh, 2021; Sari 

et al., 2023). 

  Rancang bangun sistem otomatisasi ini bertujuan untuk memberdayakan 

masyarakat agar mampu mengelola sistem hidroponik secara mandiri dan efisien. 

Dengan kontrol otomatis yang berbasis data serta monitoring menyeluruh, sistem 

ini berkontribusi dalam mendukung pertanian cerdas (smart farming) yang lebih 

berkelanjutan. Pendekatan ini tidak hanya meningkatkan hasil panen, tetapi juga 

mengurangi kesalahan manusia dalam pengelolaan nutrisi dan air, yang menjadi 

faktor krusial dalam keberhasilan pertanian hidroponik masa kini (Rahman et al., 

2020). 

1.2. Perumusan Masalah 

Permasalahan pada Laporan ini adalah : 

1. Bagaimana merancang dan membangun sistem kontrol nutrisi otomatis 

berbasis ESP32 

2. Bagaimana membangun sistem pengendalian aktuator (pompa AC dan 

pompa DC) 

3. Bagaimana merakit perangkat keras menjadi satu sistem alat yang 

terintegrasi dan dapat berfungsi stabil?  

1.3. Tujuan Penelitian 

Penelitian ini bertujuan untuk: 

1. Merancang dan membangun sistem kontrol nutrisi hidroponik 

otomatis berbasis ESP32 yang dapat bekerja secara mandiri tanpa 

intervensi manual secara terus-menerus. 

2. Mengembangkan sistem pengendalian aktuator (pompa AC dan DC) 

yang dapat dikendalikan secara otomatis dan manual, sesuai dengan 

kondisi lingkungan tanaman hidroponik. 

3. Merakit dan mengintegrasikan seluruh perangkat keras seperti sensor, 

pompa, dan modul kendali ke dalam satu sistem alat yang stabil dan 

siap digunakan di lapangan. 

1.4. Luaran 

Kegiatan ini menghasilkan beberapa luaran nyata yang dapat dimanfaatkan 

langsung oleh masyarakat, yaitu: 

1. Sebuah sistem alat kontrol nutrisi hidroponik otomatis berbasis ESP32 

yang dapat memantau dan mengatur parameter lingkungan dengan akurat. 

2. Sistem pengendalian aktuator yang bekerja responsif terhadap perubahan 

nilai sensor, serta dapat dikontrol secara otomatis dan manual. 

3. Perangkat alat yang telah terintegrasi secara fisik, bekerja stabil, dan telah 

digunakan dalam kegiatan monitoring dan pengendalian nutrisi hidroponik 

secara langsung

BAB V  

PENUTUP 

5.1 Kesimpulan 

Berdasarkan hasil perancangan, realisasi, dan pengujian yang telah 

dilakukan, dapat disimpulkan bahwa sistem hidroponik otomatis berbasis Internet 

of Things (IoT) menggunakan dua unit mikrokontroler ESP32 telah berhasil 

direalisasikan dengan baik. Sistem ini mampu membaca enam parameter utama 

lingkungan, yaitu suhu udara, kelembapan udara, suhu air, ketinggian permukaan 

air, intensitas cahaya, serta parameter nutrisi berupa nilai TDS dan pH. Seluruh 

parameter tersebut digunakan sebagai dasar dalam pengambilan keputusan 

otomatis, yang kemudian mengaktifkan aktuator berupa pompa air dan pompa 

nutrisi secara mandiri sesuai logika kontrol yang telah diprogram. 

Pengujian sistem menunjukkan bahwa seluruh perangkat keras yang 

digunakan—meliputi sensor, pompa, solenoid valve, modul relay, radar 

pelampung, power supply, serta antarmuka berupa LCD dan aplikasi Telegram—

telah terintegrasi dalam sistem secara modular dan bekerja dengan baik di 

lingkungan nyata. Sistem ini juga mendukung fleksibilitas tinggi, karena selain 

bekerja secara otomatis, pengguna dapat melakukan kontrol manual dari jarak jauh 

melalui perintah teks di Telegram. 

Dari sisi performa, sistem berhasil mencatat pembacaan data sensor 

sebanyak 10 kali dengan interval waktu tertentu. Hasil rata-rata pengukuran 

menunjukkan suhu udara sebesar 29,18 °C, kelembapan udara 80,35%, suhu air 

32,25 °C, ketinggian air 29,8 cm, nilai TDS 457,41 ppm, nilai pH 4,89, dan 

intensitas cahaya sebesar 428,7 lux. Parameter-parameter ini menunjukkan bahwa 

sistem mampu memantau kondisi lingkungan secara menyeluruh. Logika kontrol 

juga berjalan sesuai rancangan, dengan tingkat keberhasilan aktivasi pompa 

mencapai 100% selama pengujian dan waktu respons rata-rata sistem kurang dari 

satu detik antara pembacaan sensor dan pengaktifan aktuator. 

Secara keseluruhan, sistem ini telah terbukti efisien, fleksibel, dan responsif. 

Meski demikian, masih ditemukan keterbatasan pada akurasi sensor pH yang belum 

optimal akibat belum dilakukan proses kalibrasi menyeluruh. Oleh karena itu, perlu 

dilakukan penyetelan dan kalibrasi ulang pada sensor pH agar dapat berfungsi 

sebagai acuan pengambilan keputusan otomatis di masa mendatang. Dengan 

implementasi ini, sistem yang dibangun dinilai layak sebagai solusi otomasi 

hidroponik skala rumah tangga maupun urban farming secara modern dan 

berkelanjutan. 

5.2 Saran 

1. Lakukan kalibrasi sensor pH menggunakan larutan buffer standar agar nilai 

pembacaan pH lebih akurat dan dapat digunakan untuk pengendalian 

otomatis pompa pH. 

2. Tambahkan sistem penyimpanan data seperti Google Spreadsheet atau 

penyimpanan lokal untuk pencatatan histori parameter lingkungan dan 

kontrol. 

3. Gunakan enclosure panel dan pipa yang tahan air dan panas untuk 

meningkatkan daya tahan sistem di lingkungan luar ruangan. 

4. Untuk versi lanjutan, sistem dapat dikembangkan menggunakan sensor pH 

dan TDS digital I2C agar lebih stabil dan minim perawatan. 

LAMPIRAN 

 

LAMPIRAN 1 

KODE PROGRAM ESP32 SENSOR 

#include <WiFi.h> 

#include <esp_now.h> 

#include <OneWire.h> 

#include <DallasTemperature.h> 

#include <DHT.h> 

#include <Wire.h> 

#include <BH1750.h> 

#include <LiquidCrystal_I2C.h> 

#include <EEPROM.h> 

 

// Pin sensor 

#define DHTPIN 5 

#define DHTTYPE DHT22 

#define ONE_WIRE_BUS 15 

#define TRIG_PIN 18 

#define ECHO_PIN 19 

#define TDS_PIN 36 

#define PHPIN 34 

// EEPROM alamat pH kalibrasi 

const int eepromAddr = 0; 

float teganganPH4 = 0; 

 

DHT dht(DHTPIN, DHTTYPE); 

OneWire oneWire(ONE_WIRE_BUS); 

DallasTemperature sensors(&oneWire); 

BH1750 lightMeter; 

LiquidCrystal_I2C lcd(0x27, 20, 4); 

 

typedef struct struct_message { 

  float temp_dht; 

  float hum; 

  float temp_ds; 

  int jarak; 

  float tds; 

  float lux; 

  float ph; 

} struct_message; 

 

struct_message sensorData; 

 

uint8_t relayMACAddress[] = {0xF8, 0xB3, 0xB7, 0x44, 0xF4, 0xD4};  // MAC 

ESP32 relay tujuan 

// Fungsi baca jarak sensor ultrasonik 

int getDistanceCM() { 

  digitalWrite(TRIG_PIN, LOW); 

  delayMicroseconds(2); 

  digitalWrite(TRIG_PIN, HIGH); 

  delayMicroseconds(10); 

  digitalWrite(TRIG_PIN, LOW); 

  long durasi = pulseIn(ECHO_PIN, HIGH, 30000);  // timeout 30ms 

  int distance = durasi * 0.034 / 2; 

  if (distance == 0) distance = -1;  // timeout 

  return distance; 

} 

 

// Fungsi hitung pH dengan kalibrasi 

float hitungPH(float tegangan) { 

  float faktor = 3.0;  // Faktor kalibrasi, sesuaikan sesuai kalibrasi lo  

  return 4.0 - (teganganPH4 - tegangan) * faktor; 

} 

 

void onDataSent(const uint8_t *mac_addr, esp_now_send_status_t status) { 

  Serial.print("Kirim data ke MAC: "); 

  for (int i = 0; i < 6; i++) { 

    Serial.printf("%02X", mac_addr[i]); 

    if (i < 5) Serial.print(":"); 

  } 

  Serial.print(" status: "); 

  Serial.println(status == ESP_NOW_SEND_SUCCESS ? "BERHASIL" : 

"GAGAL"); 

} 

 

void setup() { 

  Serial.begin(115200); 

 

  dht.begin(); 

  sensors.begin(); 

 

  // Wire.begin();  // Optional, bisa spesifik pin jika perlu: Wire.begin(SDA, SCL); 

  Wire.begin(); 

 

  if (!lightMeter.begin()) { 

    Serial.println("Gagal inisialisasi BH1750"); 

  } 

 

  lcd.init(); 

  lcd.backlight(); 

  pinMode(TRIG_PIN, OUTPUT); 

  pinMode(ECHO_PIN, INPUT); 

 

  EEPROM.begin(32); 

  EEPROM.get(eepromAddr, teganganPH4); 

  if (isnan(teganganPH4) || teganganPH4 == 0 || teganganPH4 > 5.0) { 

    teganganPH4 = 2.5;  // default tegangan pH4 kalau EEPROM invalid 

    EEPROM.put(eepromAddr, teganganPH4); 

    EEPROM.commit(); 

    Serial.println("EEPROM kosong/invalid, set default tegangan 2.5V"); 

  } else { 

    Serial.print("Tegangan pH4 dari EEPROM: "); 

    Serial.println(teganganPH4, 3); 

  } 

 

  WiFi.mode(WIFI_STA); 

 

  if (esp_now_init() != ESP_OK) { 

    Serial.println("ESP-NOW init gagal"); 

    while (true) delay(1000); 

  } 

 

  esp_now_peer_info_t peerInfo = {}; 

  memcpy(peerInfo.peer_addr, relayMACAddress, 6); 

  peerInfo.channel = 0; 

  peerInfo.encrypt = false; 

 

  if (esp_now_add_peer(&peerInfo) != ESP_OK) { 

    Serial.println("Gagal menambahkan peer"); 

    while (true) delay(1000); 

  } 

 

  esp_now_register_send_cb(onDataSent); 

 

  Serial.println("Setup sensor selesai."); 

} 

 

void loop() { 

  // Baca sensor DHT22 

  float tDHT = dht.readTemperature(); 

  float hDHT = dht.readHumidity(); 

  if (isnan(tDHT) || isnan(hDHT)) { 

    Serial.println("Gagal baca sensor DHT22"); 

    sensorData.temp_dht = -999; 

    sensorData.hum = -999; 

  } else { 

    sensorData.temp_dht = tDHT; 

    sensorData.hum = hDHT; 

  } 

 

  // Baca sensor DS18B20 

  sensors.requestTemperatures(); 

  float tempDS = sensors.getTempCByIndex(0); 

  if (tempDS == DEVICE_DISCONNECTED_C) { 

    Serial.println("Gagal baca sensor DS18B20"); 

    sensorData.temp_ds = -999; 

  } else { 

    sensorData.temp_ds = tempDS; 

  } 

 

  // Baca jarak ultrasonic 

  int jarak = getDistanceCM(); 

  sensorData.jarak = jarak; 

 

  // Baca sensor TDS dengan kalibrasi 

  int adcTDS = analogRead(TDS_PIN); 

  float voltTDS = adcTDS * 3.3 / 4095.0; 

  sensorData.tds = (133.42 * voltTDS * voltTDS - 255.86 * voltTDS + 857.39) * 

1.21; 

  // Baca sensor BH1750 (cahaya) 

  float lux = lightMeter.readLightLevel(); 

  if (lux < 0) { 

    Serial.println("Gagal baca sensor BH1750"); 

    sensorData.lux = 0; 

  } else { 

    sensorData.lux = lux; 

  } 

 

  // Baca sensor pH dengan kalibrasi 

  int adcPH = analogRead(PHPIN); 

  float tegPH = (adcPH / 4095.0) * 3.3; 

  sensorData.ph = hitungPH(tegPH); 

 

  // Debug output ke Serial 

  Serial.printf("s.udara: %.2f°C, %.2f%% | s.air: %.2f°C | Jarak: %d cm | TDS: %.2f 

ppm | Lux: %.2f lx | pH: %.2f\n", 

                sensorData.temp_dht, sensorData.hum, sensorData.temp_ds, 

                sensorData.jarak, sensorData.tds, sensorData.lux, sensorData.ph); 

 

  // Tampilkan ke LCD I2C 

  lcd.clear(); 

  lcd.setCursor(0, 0); 

  lcd.printf("s.ling:%.1fC H:%.0f%%", sensorData.temp_dht, sensorData.hum); 

  lcd.setCursor(0, 1); 

  if (sensorData.jarak == -1) { 

    lcd.printf("s.air:%.1fC J:--cm", sensorData.temp_ds); 

  } else { 

    lcd.printf("s.air:%.1fC J:%dcm", sensorData.temp_ds, sensorData.jarak); 

  } 

  lcd.setCursor(0, 2); 

  lcd.printf("TDS:%.0fppm pH:%.2f", sensorData.tds, sensorData.ph); 

  lcd.setCursor(0, 3); 

  lcd.printf("Lux:%.0flx", sensorData.lux); 

 

  // Kirim data via ESP-NOW ke relay 

  esp_err_t result = esp_now_send(relayMACAddress, (uint8_t *)&sensorData, 

sizeof(sensorData)); 

  if (result == ESP_OK) { 

    Serial.println("Data terkirim ke relay."); 

  } else { 

    Serial.println("Gagal kirim data."); 

  } 

 

  delay(3000);  // Delay 3 detik, bisa diganti millis() kalau mau multitasking 

} 

KODE PROGRAM ESP32 RELAY 

#include <WiFi.h> 

#include <esp_now.h> 

#include <WiFiClientSecure.h> 

#include <UniversalTelegramBot.h> 

 

// ===== WIFI & TELEGRAM ===== 

const char* ssid = "POSYANDU MAKMUR"; 

const char* password = "Kaderrw04"; 

#define BOT_TOKEN "7438614388:AAE9r4E1UQbJZsu7sTHHQhoCzMAUQf-

x3TU" 

 

WiFiClientSecure client; 

UniversalTelegramBot bot(BOT_TOKEN, client); 

unsigned long lastCheck = 0; 

 

// ===== DATA SENSOR dari ESP-NOW ===== 

typedef struct SensorData { 

  float temp_dht; 

  float hum; 

  float temp_ds; 

  int jarak; 

  float tds; 

  float lux; 

  float ph; 

} SensorData; 

SensorData sensorData; 

 

// ===== PIN OUTPUT ===== 

#define M1 4     // Pompa Air 

#define V1 23    // Nutrisi A 

#define V2 13    // Nutrisi B 

#define V3 14    // Pompa pH 

 

// ===== VARIABEL OTOMATIS AB MIX ===== 

unsigned long startNutrisiTime = 0; 

unsigned long waktuPantauAB = 0; 

bool nutrisiSedangNyala = false; 

bool sedangPantauAB = false; 

 

// ===== MODE & MANUAL STATE ===== 

bool modeOtomatis = true; 

bool manual_M1 = false; 

bool manual_V1V2 = false; 

bool manual_V3 = false; 

bool sudahMulai = false; 

 

void setup() { 

  Serial.begin(115200); 

  WiFi.mode(WIFI_STA); 

  WiFi.begin(ssid, password); 

  client.setInsecure(); 

 

  while (WiFi.status() != WL_CONNECTED) { 

    delay(500); 

    Serial.print("."); 

  } 

  Serial.println("\n✅ WiFi Tersambung!"); 

 

  int allPins[] = {M1, V1, V2, V3}; 

  for (int i = 0; i < 4; i++) { 

    pinMode(allPins[i], OUTPUT); 

    digitalWrite(allPins[i], LOW); 

  } 

 

  if (esp_now_init() != ESP_OK) { 

    Serial.println("❌ Gagal init ESP-NOW"); 

  } else { 

    esp_now_register_recv_cb(OnDataRecv); 

    Serial.println("✅ ESP32 Relay siap menerima data"); 

  } 

} 

void OnDataRecv(const esp_now_recv_info_t *info, const uint8_t *incomingData, 

int len) { 

  memcpy(&sensorData, incomingData, sizeof(sensorData)); 

  Serial.println(" Data diterima dari ESP32 Sensor:"); 

  Serial.printf("DHT: %.1f°C, %.1f%% | DS18B20: %.1f°C\n", 

sensorData.temp_dht, sensorData.hum, sensorData.temp_ds); 

  Serial.printf("Jarak: %d cm | TDS: %.1f ppm | Lux: %.0f lx | pH: %.2f\n", 

sensorData.jarak, sensorData.tds, sensorData.lux, sensorData.ph); 

 

  if (modeOtomatis) kontrolPompa(); 

} 

 

void kontrolPompa() { 

  Serial.println(" Mode: OTOMATIS"); 

 

  // === POMPA AIR === 

  if (sensorData.jarak > 25 || sensorData.tds > 1200) { 

    digitalWrite(M1, HIGH); 

    Serial.println(" Pompa AIR ON"); 

  } else { 

    digitalWrite(M1, LOW); 

    Serial.println(" Pompa AIR OFF"); 

  } 

  // === AB MIX LOGIC === 

  if (sensorData.tds < 900 && !nutrisiSedangNyala && !sedangPantauAB) { 

    digitalWrite(V1, HIGH); 

    digitalWrite(V2, HIGH); 

    startNutrisiTime = millis(); 

    nutrisiSedangNyala = true; 

    Serial.println(" Pompa AB MIX ON selama 45 detik"); 

  } 

 

  if (nutrisiSedangNyala && millis() - startNutrisiTime >= 45000) { 

    digitalWrite(V1, LOW); 

    digitalWrite(V2, LOW); 

    nutrisiSedangNyala = false; 

    sedangPantauAB = true; 

    waktuPantauAB = millis(); 

    Serial.println("⏳ Pompa AB OFF, pantau TDS selama 15 detik"); 

  } 

 

  if (sedangPantauAB && millis() - waktuPantauAB >= 15000) { 

    if (sensorData.tds < 900) { 

      digitalWrite(V1, HIGH); 

      digitalWrite(V2, HIGH); 

      startNutrisiTime = millis(); 

      nutrisiSedangNyala = true; 

      sedangPantauAB = false; 

      Serial.println(" TDS masih < 900 ppm, ulangi pompa AB MIX 45 detik"); 

    } else { 

      sedangPantauAB = false; 

      Serial.println("✅ TDS sudah cukup, pompa AB berhenti"); 

    } 

  } 

 

  // === POMPA PH === 

  if (sensorData.ph < 4) { 

    digitalWrite(V3, HIGH); 

    Serial.println("⚗ Pompa pH ON"); 

  } else { 

    digitalWrite(V3, LOW); 

    Serial.println("⚗ Pompa pH OFF"); 

  } 

} 

 

void kontrolManual() { 

  digitalWrite(M1, manual_M1 ? HIGH : LOW); 

  digitalWrite(V1, manual_V1V2 ? HIGH : LOW); 

  digitalWrite(V2, manual_V1V2 ? HIGH : LOW); 

  digitalWrite(V3, manual_V3 ? HIGH : LOW); 

} 

 

void loop() { 

  if (!modeOtomatis) kontrolManual(); 

 

  if (millis() - lastCheck > 2000) { 

    int numNew = bot.getUpdates(bot.last_message_received + 1); 

    while (numNew) { 

      handleNewMessages(numNew); 

      numNew = bot.getUpdates(bot.last_message_received + 1); 

    } 

    lastCheck = millis(); 

  } 

 

  delay(100); 

} 

 

void handleNewMessages(int numNewMessages) { 

  for (int i = 0; i < numNewMessages; i++) { 

    String text = bot.messages[i].text; 

    String chat_id = bot.messages[i].chat_id; 

    Serial.println(" Telegram CMD: " + text); 

 

    if (text == "/mulai" || text == "/start") { 

      sudahMulai = true; 

      String welcomeMsg = 

        " Selamat datang di Sistem Hidroponik POSYANDU RW4!\n\n" 

        " Perintah:\n" 

        "/auto - Mode otomatis\n" 

        "/manual - Mode manual\n\n" 

        " Manual:\n" 

        "/POMPA_AIR_ON /OFF\n" 

        "/POMPA_AB_ON /OFF\n" 

        "/POMPA_PH_ON /OFF\n\n" 

        " /status - Lihat data"; 

      bot.sendMessage(chat_id, welcomeMsg, ""); 

      return; 

    } 

 

    if (!sudahMulai) { 

      bot.sendMessage(chat_id, "⛔ Ketik /mulai dulu!", ""); 

      return; 

    } 

    if (text == "/auto") { 

      modeOtomatis = true; 

      bot.sendMessage(chat_id, " Mode otomatis diaktifkan.", ""); 

    } else if (text == "/manual") { 

      modeOtomatis = false; 

      bot.sendMessage(chat_id, " Mode manual diaktifkan.", ""); 

    } 

 

    if (!modeOtomatis) { 

      if (text == "/POMPA_AIR_ON") { 

        manual_M1 = true; 

        kontrolManual(); 

        bot.sendMessage(chat_id, " Pompa Air: ON ✅", ""); 

      } else if (text == "/POMPA_AIR_OFF") { 

        manual_M1 = false; 

        kontrolManual(); 

        bot.sendMessage(chat_id, " Pompa Air: OFF ❌", ""); 

      } else if (text == "/POMPA_AB_ON") { 

        manual_V1V2 = true; 

        kontrolManual(); 

        bot.sendMessage(chat_id, " Pompa Nutrisi A&B: ON ✅", ""); 

      } else if (text == "/POMPA_AB_OFF") { 

        manual_V1V2 = false; 

        kontrolManual(); 

        bot.sendMessage(chat_id, " Pompa Nutrisi A&B: OFF ❌", ""); 

      } else if (text == "/POMPA_PH_ON") { 

        manual_V3 = true; 

        kontrolManual(); 

        bot.sendMessage(chat_id, "⚗ Pompa pH: ON ✅", ""); 

      } else if (text == "/POMPA_PH_OFF") { 

        manual_V3 = false; 

        kontrolManual(); 

        bot.sendMessage(chat_id, "⚗ Pompa pH: OFF ❌", ""); 

      } 

    } 

 

    if (text == "/status") { 

      String msg = " Status Saat Ini:\n"; 

      msg += "Mode: " + String(modeOtomatis ? "OTOMATIS" : "MANUAL") + 

"\n"; 

      msg += "\n Aktuator:\n"; 

      msg += "Pompa Air: " + String(digitalRead(M1) ? "ON\n" : "OFF\n"); 

      msg += "Nutrisi A: " + String(digitalRead(V1) ? "ON\n" : "OFF\n"); 

      msg += "Nutrisi B: " + String(digitalRead(V2) ? "ON\n" : "OFF\n"); 

      msg += "Pompa pH: " + String(digitalRead(V3) ? "ON\n" : "OFF\n"); 

      msg += "\n Sensor:\n"; 

      msg += "DHT22: " + String(sensorData.temp_dht, 1) + "°C, " + 

String(sensorData.hum, 1) + "%\n"; 

      msg += "DS18B20: " + String(sensorData.temp_ds, 1) + "°C\n"; 

      msg += "Jarak: " + String(sensorData.jarak) + " cm\n"; 

      msg += "TDS: " + String(sensorData.tds, 1) + " ppm\n"; 

      msg += "Lux: " + String(sensorData.lux, 0) + " lx\n"; 

      msg += "pH: " + String(sensorData.ph, 2); 

      bot.sendMessage(chat_id, msg, ""); 

    } 

  } 

} 

LAMPIRAN 2 

